

In-class Demonstration: Quadratic Forms; Orthogonal Diagonalization

By Russell Blyth, modified by Mike May, S.J.

```
> restart: with(LinearAlgebra): with(plots): with(plottools):  
Warning, the name changecoords has been redefined  
Warning, the assigned name arrow now has a global binding
```

Example 1: The conic $-x^2 - y^2 + 6xy = 1$

A is the matrix of the quadratic form

```
> A := Matrix(2,2, [[-1,3],[3,-1]]);
```

$$A := \begin{bmatrix} -1 & 3 \\ 3 & -1 \end{bmatrix} \quad (1)$$

Find the eigenvalues and eigenvectors of A:

```
> evA := [Eigenvectors(A)];
```

$$evA := \left[\begin{bmatrix} 2 \\ -4 \end{bmatrix}, \begin{bmatrix} 1 & -1 \\ 1 & 1 \end{bmatrix} \right] \quad (2)$$

Normalize the eigenvectors:

```
> v1 := Normalize(Column(evA[2],1),Euclidean);  
v2 := Normalize(Column(evA[2],2),Euclidean);
```

$$v1 := \begin{bmatrix} \frac{1}{2}\sqrt{2} \\ \frac{1}{2}\sqrt{2} \end{bmatrix}$$
$$v2 := \begin{bmatrix} -\frac{1}{2}\sqrt{2} \\ \frac{1}{2}\sqrt{2} \end{bmatrix} \quad (3)$$

Construct a matrix P which orthogonally diagonalizes A

```
> PA := <v1|v2>;
```

$$PA := \begin{bmatrix} \frac{1}{2}\sqrt{2} & -\frac{1}{2}\sqrt{2} \\ \frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} \end{bmatrix} \quad (4)$$

Compute the diagonal matrix (since $A = PDP^t$, we have $D = P^tAP$)

```
> DA := Transpose(PA) . A . PA;
```

$$DA := \begin{bmatrix} 2 & 0 \\ 0 & -4 \end{bmatrix} \quad (5)$$

Make Maple compute the equation of the rotated conic:

```
> LHSA := Transpose(<X,Y>) . DA . <X,Y>;
```


$$LHSA := 2X^2 - 4Y^2$$

(6)

The equation is $LHSA = 1$, which is the equation of a hyperbola.


Let's graph; first the conic relative to the new axes:

```
> implicitplot(-4*x^2+2*y^2=1,x=-2..2,y=-2..2,scaling=constrained);
```



and then the graph relative to the original axes:

```
> v1g := line([0,0],convert(v1,list),color=red,thickness=2):  
v2g := line([0,0],convert(v2,list),color=red,thickness=2):  
conic := implicitplot(-x^2-y^2+6*x*y=1,x=-2..2,y=-2..2):  
display([v1g,v2g,conic],scaling=constrained);
```



Subtle point: the axes have been reversed here - to prevent this we should reverse the order we choose v_1 and v_2 - in general, choose the order of the columns of P so that $\det(P) = 1$.

Example 2: The quadric surface $2x^2 + 3y^2 + 23z^2 + 72xz + 150 = 0$

B is the matrix of the quadratic form

```
> B := Matrix(3,3,[[2,0,36],[0,3,0],[36,0,23]]);
```

$$B := \begin{bmatrix} 2 & 0 & 36 \\ 0 & 3 & 0 \\ 36 & 0 & 23 \end{bmatrix} \quad (7)$$

Find the eigenvalues and eigenvectors of B

```
> evB := [Eigenvectors(B)];
```

(8)

$$evB := \left[\begin{array}{c} \left[\begin{array}{c} 50 \\ -25 \\ 3 \end{array} \right], \left[\begin{array}{ccc} \frac{3}{4} & \frac{-4}{3} & 0 \\ 0 & 0 & 1 \\ 1 & 1 & 0 \end{array} \right] \end{array} \right] \quad (8)$$

Normalize the eigenvectors:

```
> w1 := Normalize(Column(evB[2],1),Euclidean);
w2 := Normalize(Column(evB[2],2),Euclidean);
w3 := Normalize(Column(evB[2],3),Euclidean);
```

$$w1 := \begin{bmatrix} \frac{3}{5} \\ \frac{4}{5} \\ 0 \end{bmatrix}$$

$$w2 := \begin{bmatrix} \frac{-4}{5} \\ 0 \\ \frac{3}{5} \end{bmatrix}$$

$$w3 := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix} \quad (9)$$

Construct a matrix P which orthogonally diagonalizes A:

```
> PB := <w1|w2|w3>;
```

$$PB := \begin{bmatrix} \frac{3}{5} & \frac{-4}{5} & 0 \\ 0 & 0 & 1 \\ \frac{4}{5} & \frac{3}{5} & 0 \end{bmatrix} \quad (10)$$

Compute the diagonal matrix (since $A = PDP^t$, we have $D = P^tAP$)

```
> DB := Transpose(PB) . B . PB;
```

$$DB := \begin{bmatrix} 50 & 0 & 0 \\ 0 & -25 & 0 \\ 0 & 0 & 3 \end{bmatrix} \quad (11)$$

Make Maple compute the equation of the rotated conic:

```
> LHSB := Transpose(<X,Y,Z>) . DB . <X,Y,Z>;
```


$$LHSB := 50 X^2 - 25 Y^2 + 3 Z^2 \quad (12)$$

>

The equation is $LHSB = -150$, which is the equation of a paraboloid of two sheets.

Let's graph, first the quadric surface relative to the new axes:

```
> implicitplot3d(3*x^2+50*y^2-25*z^2=-150,x=-10..10,y=-10..10,z=-10..10,axes=normal,scaling=constrained,grid=[20,20,20]);
```



And then relative to the original axes: (the eigenspace basis vectors have been stretched to make them long enough to see clearly)

```
> w15 := evalm(5*w1); w25 := evalm(5*w2); w35 := evalm(5*w3);
```

```
      w15 := [ 3  0  4 ]
```

```
      w25 := [-4  0  3 ]
```

```
      w35 := [ 0  5  0 ]
```

(13)

```
> w1g := line([0,0,0],convert(w15,list),color=red,thickness=4):  
w2g := line([0,0,0],convert(w25,list),color=red,thickness=4):  
w3g := line([0,0,0],convert(w35,list),color=red,thickness=4):  
conicB := implicitplot3d(2*x^2 + 3*y^2 + 23*z^2 + 72*x*z + 150 =  
0, x=-10..10,y=-10..10,z=-10..10,axes=normal,grid=[20,20,20]):  
display([w1g,w2g,w3g,conicB],scaling=constrained);
```

